1 题目

Let e_n the sequence of integers such as the decimal development of 2^{en} starts with the following digits: 66666…

For instance we have:
$e_1 = 224296 \rightarrow 2224296 = 666664682427419933022…$
$e_5 = 620220 \rightarrow 2620220 = 666669696065907296336…$

What is $e_{1000000}$?

2 解析

解法 1

暴力搜索序列 e 前 20 项如下：
`224296, 295073, 478666, 549443, 620220, 803813, 874590, 945367, 1128960, 1199737, 1454107, 1524884, 1779254, 1850031, 2104401, 2175178, 2429548, 2500325, 2754695, 2825472, ...`
相邻两项做差得新序列如下：
`70777, 183593, 70777, 70777, 183593, 70777, 70777, 183593, 70777, 254370, 70777, 254370, 70777, 254370, 70777, 254370, 70777, 254370, 70777, ...`
发现新序列只有 3 种不同取值`{70777, 183593, 254370}`（原理未知）。
可以从 $e_1 = 224296$ 开始，每次尝试`{70777, 183593, 254370}`三种步进，选择满足条件的一种步进，更新序列 e 的值，直到找到 $e_{1000000}$。
代码略。

解法 2

设 e 为满足条件的正整数，则

$$66666 \times 10^k \le 2^e < 66667 \times 10^k$$

其中 k 为非负整数，两边以 2 为底取对数得

$$\begin{aligned}
&(\log_2 66666) + k(\log_2 10) \le e < (\log_2 66667) + k(\log_2 10) \\
&\Leftrightarrow a_1 + m \cdot k \le e < a_2 + m \cdot k \\
&\Rightarrow e - a_2 < m \cdot k \le e - a_1 \\
&\Rightarrow \{e - a_2\} < \{m \cdot k\} \le \{e - a_1\}
\end{aligned}$$

这里$\{e\text{-}a_2\}$, $\{m\,k\}$, $\{e\text{-}a_1\}$取值均在$[0, 1)$范围。
实际上我们需要求出非负整数 k 使得$\{m\,k\}$落在一个宽度为 $d = a_2\text{-}a_1$ 的小区间内。
使用连分式展开逼近算法可以求出无理数 m 的有理近似值 p/q，满足

$$\left| m - \frac{p}{q} \right| \le \frac{1}{q^2}$$

取适当大小的阈值 $q_{min} = N/d$，$q > q_{min}$，使得可行解数大于 N。将$\{e\text{-}a_2\}$和$\{e\text{-}a_1\}$用分母 q 的有理数近似，则不等式等价于

$$\left\{\frac{p_2}{q}\right\} < \left\{\frac{k\cdot p}{q}\right\} \le \left\{\frac{p_1}{q}\right\}$$
$$\Leftrightarrow (p_2 \bmod q) < (k\cdot p \bmod q) \le (p_1 \bmod q)$$

令非负整数 $x = k\,p \bmod q$，则 x 可能取值范围为$[p_2 \bmod q - \varepsilon, p_1 \bmod q + \varepsilon]$，考虑有理近似误差，范围增加一点余量。从而$k = x\cdot p^{-1} \bmod q$（因 p 和 q 互素，模逆一定存在），再通过检验

$$\lceil a_1 + m\cdot k\rceil = \lfloor a_2 + m\cdot k\rfloor = e$$

确认 k 为有效解，所求 e 为序列中某一项。
将搜索出的所有 e 值从小到大排序求出第 1000000 项即 $e_{1000000}$。
代码见小节 3。

3　代码

解法 2 的 Pari/GP 代码如下：

```
default(parisize, 3*10^8);

{
N = 1000000;

a1 = log(66666)/log(2);
a2 = log(66667)/log(2);
m = log(10)/log(2);
d = a2 - a1;

qmin = floor(N/d*1.001); B = qmin;
while (1,
    rat = bestappr(m, B);
    p = numerator(rat);
    q = denominator(rat);
    if (q > qmin, break, B = floor(B*1.1));
);
pa2 = round(frac(a2)*q);
pd = round(d*q);

ve = List();
for (x = -2, pd+2,
    k = lift(Mod(x-pa2, q) / Mod(p, q));
```

```
    if (ceil(a1+k*m) == floor(a2+k*m),
        listput(ve, floor(a2+k*m));
    );
);
ve = vecsort(Vec(ve));
printf("e(%d) = %d\n", N, ve[N]);
}
```